HRP 261 SAS LAB TWO January 25, 2012

Lab Two: Mantel-Haenszel test and Mantel-Haenszel OR/RR, introduction to McNemar’s Test

Lab Objectives

After today’s lab you should be able to:

1. Distinguish between numeric and character variables in SAS.

2. Use PROC FREQ to generate stratified 2x2 tables. 

3. Use PROC FREQ to generate Mantel-Haenszel statistics and the Mantel-Haenszel summary OR/RR and Breslow-Day test of Homogeneity.

4. Interpret Mantel-Haenszel results.

5. Practice using the RESULTS browser (left hand side of screen) to scroll through output.

6. Input paired data directly into SAS.

7. Use PROC FREQ to generate McNemar’s statistic for paired or matched data.

8. Change grouped data into a dataset that contains 1 observation for each individual.

9. Use a do-loop.

10. Use an if-then-do loop.

11. Use nested loops.

12. Use the output statement to add observations to a dataset that you are modifying.

13. See the connection between CMH and McNemar’s tests.

SAS PROCs                                     

SAS EG equivalent

PROC FREQ                             

Describe(Table Analysis
PROC LOGISTIC



Analyze(Regresssion(Logistic regression
 LAB EXERCISE STEPS:

Follow along with the computer in front…

1. Double-click on the SAS EG icon to open SAS EG.

2. Select New Project.

3. On the menus, select File>New>Data
4. Create a new dataset admissions in the WORK library.
In Name, type admissions (SAS is cap-insensitive). Then click on the work library (filing cabinet drawers represent libraries). 

.
Then Click Next>
Click on the first variable (A). Name the variable “program” and then as save it as character variable (default).

Call the subsequent variables: IsFemale, Denied, Count and save them as numeric variables. Click on Finish to look at your data.

Your data table will behave just as an excel table. First select with the last two columns and delete them with the key Delete (just as if it would be an excel table!)
Go to the class website: www.stanford.edu/~kcobb/courses/hrp261 --> Lab 2 Data

Highlight the following data with your mouse; copy with ctrl C:

[image: image5.png]o)\ H9- )+ Data_lab2 - Microsoft Excel -5 x
W)
Home  Inset  Pagelayout  Formulas  Dats | Review  View  AddIns  Acrobat @ - 7 x
FEY ﬁ YE ﬁ = Connedtions D & Clear E BrE @ng # Show Detail
) 3 : en = g1 = ]
) 5 B e i G Resprly & =2 e Detan
o G G G| R 3} sou | e 3 e e el e
Access Web  Text Sources~ | Connections All~ 52 Edit Links 7 Advanced | Columns Duplicates Validation ~ Analysis ~ - -
G o S o .
Al M £ A 2
E 1 f 6 [ i 3 T t I m [~ ol »l alr s

25|
2|
27|
2|
2|
30|
31
32|
33
34|
35|
Ed
37|
38
W4 ¥ W] Sheet1 Sheet2 ”Sheet3 ¥ A L
Reay |

Average: 630555555 _ Countis6 _ Sum: 540 |[25 00 00— U 8)
1

& SASEmeprezGude | (i




Paste it with Ctrl-V. Note that the numeric variables are aligned toward the right, but the character variable is aligned towards the left. [image: image6.png]Fle Edt View Tasks Progam  Tods Hep |E-5- @& 92 B3 X | ® o |- [EegProcess Flow ~

5. It is much easier to create the dataset by programming. The code to create the data set in SAS is the following: 
data admissions;

input program $ IsFemale   Denied   Count;

datalines;

A 1 1  19

A 1 0  89

A 0 1  314

A 0 0  511

B 1 1  8

B 1 0  17

B 0 1  208

B 0 0  352

C 1 1  391

C 1 0  202

C 0 1  205

C 0 0  120

D 1 1  248

D 1 0  127

D 0 1  265

D 0 0  142

E 1 1  289

E 1 0  104

E 0 1  147

E 0 0  44

F 1 1  321

F 1 0  20

F 0 1  347

F 0 0  26

;

run;
6. Now we will test for association between gender and admissions. First we have to create a 2x2 table. Click on Describe>Table Analysis

In the Data Screen drag “IsFemale” and “Denied” to make them the Table variables and “count” to make it the frequency (weighting) variable.

Click on Tables in the left-hand menu. In the Tables screen, drag and drop IsFemale to make it the row variable, and drag Denied to make it the column variable.

In the Cell Statistics menu check the boxes Cell Frequencies and Expected Cell Frequency (uncheck Cell percentages if it is checked)

Click on Table Statistics>Association in the left-hand menu. In this screen check the boxes labeled Chi-Square tests, Fisher’s exact test and Measures. 

Click on Preview Code, which allows you to see the code that SAS has automatically generated.

The relevant code that generates the table and the statistics is the PROC FREQ (the frequency procedure) code. Some of the automatically generated code is not essential. Here is the code I would use to repeat the analyses that we have just done with point-and-click:

/**IS gender related to denial of graduate admissions at Berkeley?**/
proc freq data=admissions order=data; 

tables isfemale*denied  /chisq nocol nopercent norow measures expected ;

weight count;

run;

When doing the homeworks, you may decide that it is faster to use coding rather than point-and-click, so save the code for future reference!

If you are familiar with the code, you can also directly modify the automatically generated code, which can save you repeated pointing and clicking. I will show an example in class.

Review the code (as a class). Then close the code and click Run to run the code. 

OUTPUT:



	Table of IsFamale by Denied

	 
	Denied
	Total

	
	1
	0
	

	IsFamale
	 
	1276
	559
	1835

	1
	Frequency
	
	
	

	
	Expected
	1122.3
	712.71
	 

	0
	Frequency
	1486
	1195
	2681

	
	Expected
	1639.7
	1041.3
	 

	Total
	Frequency
	2762
	1754
	4516


	Statistic
	DF
	Value
	Prob

	Chi-Square
	1
	91.2997
	<.0001

	Likelihood Ratio Chi-Square
	1
	92.5105
	<.0001

	Continuity Adj. Chi-Square
	1
	90.7067
	<.0001

	Mantel-Haenszel Chi-Square
	1
	91.2795
	<.0001

	Phi Coefficient
	 
	0.1422
	 

	Contingency Coefficient
	 
	0.1408
	 

	Cramer's V
	 
	0.1422
	 



	Fisher's Exact Test

	Cell (1,1) Frequency (F)
	1276

	Left-sided Pr <= F
	1.0000

	Right-sided Pr >= F
	4.579E-22

	 
	 

	Table Probability (P)
	3.846E-22

	Two-sided Pr <= P
	8.408E-22


	Estimates of the Relative Risk (Row1/Row2)

	Type of Study
	Value
	95% Confidence Limits

	Case-Control (Odds Ratio)
	1.8356
	1.6196
	2.0805

	Cohort (Col1 Risk)
	1.2546
	1.1988
	1.3130

	Cohort (Col2 Risk)
	0.6834
	0.6303
	0.7411




7. Could Program be a confounder of the relationship between gender and denial of admissions?
Is program related to gender?

Is program related to admissions rates?

We can repeat the Chi-square test for the variables IsFemale and Program
Describe>Table

Select IsFemale and Progam as Table variables. Select Count as Frequency count

Go to Tables in the left hand side and Drag IsFemale and Program

On the left hand menu select Association. Then click on the boxes Chi-square tests and Measures 

Then Click Run.
FYI, the corresponding codes for this analysis is:
proc freq data=admissions order=data; 

tables program*isfemale  /chisq nocol nopercent norow measures ;

weight count;

run;

OUTPUT:
	Table of program by IsFamale

	 
	IsFamale
	Total

	
	1
	0
	

	program
	 
	108
	825
	933

	A
	Frequency
	
	
	

	B
	Frequency
	25
	560
	585

	C
	Frequency
	593
	325
	918

	D
	Frequency
	375
	407
	782

	E
	Frequency
	393
	191
	584

	F
	Frequency
	341
	373
	714

	Total
	Frequency
	1835
	2681
	4516


	Statistic
	DF
	Value
	Prob

	Chi-Square
	5
	1070.2064
	<.0001

	Likelihood Ratio Chi-Square
	5
	1223.1456
	<.0001

	Mantel-Haenszel Chi-Square
	1
	508.8900
	<.0001

	Phi Coefficient
	 
	0.4868
	 

	Contingency Coefficient
	 
	0.4377
	 

	Cramer's V
	 
	0.4868
	 


8. Repeat the analysis for Program and Denied using code. New(Program:.
proc freq data=admissions order=data; 

tables program*denied  /chisq nocol nopercent norow measures ;

weight count;

run;
	Table of program by Denied

	 
	Denied
	Total

	
	1
	0
	

	program
	 
	333
	600
	933

	A
	Frequency
	
	
	

	B
	Frequency
	216
	369
	585

	C
	Frequency
	596
	322
	918

	D
	Frequency
	513
	269
	782

	E
	Frequency
	436
	148
	584

	F
	Frequency
	668
	46
	714

	Total
	Frequency
	2762
	1754
	4516


	Statistic
	DF
	Value
	Prob

	Chi-Square
	5
	771.6742
	<.0001

	Likelihood Ratio Chi-Square
	5
	848.5218
	<.0001

	Mantel-Haenszel Chi-Square
	1
	717.1111
	<.0001

	Phi Coefficient
	 
	0.4134
	 

	Contingency Coefficient
	 
	0.3820
	 

	Cramer's V
	 
	0.4134
	 


So, Program is strongly related to both gender and admissions 

9. Now, stratify on Program…
Click on the Admissions dataset. Click on Describe>Tables. Then Select IsFemale, Denied and Program as Table Variables and Count as Frequency count.

On the left hand menu select Cell Statistics. Then check Cell Frequencies and Expected cell Frequency

Drag all the variables in the Tables Menu on the left-hand side, as pictured below (hint: drag denied in first, then isfemale, then program):

On the left hand menu select Association. Then check  the Chi-Square tests and measures boxes.

Then Click Run.
FYI, the corresponding code would be:
/**IS gender related to denial of graduate admissions at Berkeley after adjusting for program?**/
proc freq data=admissions order=data; 

tables program*isfemale*denied  /cmh nocol nopercent norow measures;

weight count;

run; 

OUTPUT:
	Table 1 of IsFamale by Denied

	Controlling for program=A

	 
	Denied
	Total

	
	0
	1
	

	IsFamale
	 
	511
	314
	825

	0
	Frequency
	
	
	

	
	Col Pct
	85.17
	94.29
	 

	1
	Frequency
	89
	19
	108

	
	Col Pct
	14.83
	5.71
	 

	Total
	Frequency
	600
	333
	933


	Estimates of the Relative Risk (Row1/Row2)

	Type of Study
	Value
	95% Confidence Limits

	Case-Control (Odds Ratio)
	0.3474
	0.2076
	0.5814

	Cohort (Col1 Risk)
	0.7516
	0.6786
	0.8325

	Cohort (Col2 Risk)
	2.1634
	1.4252
	3.2840



	Table 2 of IsFamale by Denied

	Controlling for program=B

	 
	Denied
	Total

	
	0
	1
	

	IsFamale
	 
	352
	208
	560

	0
	Frequency
	
	
	

	
	Col Pct
	95.39
	96.30
	 

	1
	Frequency
	17
	8
	25

	
	Col Pct
	4.61
	3.70
	 

	Total
	Frequency
	369
	216
	585


	Estimates of the Relative Risk (Row1/Row2)

	Type of Study
	Value
	95% Confidence Limits

	Case-Control (Odds Ratio)
	0.7964
	0.3378
	1.8775

	Cohort (Col1 Risk)
	0.9244
	0.7012
	1.2186

	Cohort (Col2 Risk)
	1.1607
	0.6489
	2.0762


                                    Controlling for program=C

	Table 3 of IsFamale by Denied

	Controlling for program=C

	 
	Denied
	Total

	
	0
	1
	

	IsFamale
	 
	120
	205
	325

	0
	Frequency
	
	
	

	
	Col Pct
	37.27
	34.40
	 

	1
	Frequency
	202
	391
	593

	
	Col Pct
	62.73
	65.60
	 

	Total
	Frequency
	322
	596
	918


	Estimates of the Relative Risk (Row1/Row2)

	Type of Study
	Value
	95% Confidence Limits

	Case-Control (Odds Ratio)
	1.1331
	0.8545
	1.5024

	Cohort (Col1 Risk)
	1.0839
	0.9045
	1.2989

	Cohort (Col2 Risk)
	0.9566
	0.8645
	1.0586


ETC...

	Cochran-Mantel-Haenszel Statistics (Based on Table Scores)

	Statistic
	Alternative Hypothesis
	DF
	Value
	Prob

	1
	Nonzero Correlation
	1
	1.4972
	0.2211

	2
	Row Mean Scores Differ
	1
	1.4972
	0.2211

	3
	General Association
	1
	1.4972
	0.2211


	Estimates of the Common Relative Risk (Row1/Row2)

	Type of Study
	Method
	Value
	95% Confidence Limits

	Case-Control
	Mantel-Haenszel
	0.9049
	0.7717
	1.0612

	(Odds Ratio)
	Logit
	0.9284
	0.7894
	1.0918

	Cohort
	Mantel-Haenszel
	0.9451
	0.8660
	1.0314

	(Col1 Risk)
	Logit
	0.8652
	0.8030
	0.9322

	Cohort
	Mantel-Haenszel
	1.0275
	0.9830
	1.0739

	(Col2 Risk)
	Logit
	0.9958
	0.9645
	1.0280



	Breslow-Day Test for
Homogeneity of the Odds Ratios

	Chi-Square
	18.5989

	DF
	5

	Pr > ChiSq
	0.0023


10. Just to show the parallels (and as a preview of what’s to come), let’s also run a multivariate regression (logistic regression) with and without adjustment for confounding by program to see what effect that has on the point estimate for gender. Don’t worry too much about the coding right now—we’ll get back to this later in the term. 
In the Input Data window select Analyze>>Regression>>Logistic Regression

On the left menu select Data. Then Selcet Denied as dependent variable, IsFemale as quantitative variable and Count as Relative weight

On the left hand menu select Response. Then select Fit model to level 1. This will model the odds of being denied (Denied=1), rather than the odds of being accepted. 

On the Left hand Menu select Effects. Then click on IsFemale and Select Main. Finally click on Run.

This is a simple logistic model with gender as the only predictor. Corresponding SAS code:

proc logistic data=admissions;

model denied (event="1") = isfemale;

weight count;

run;



Here is the parameter estimate (beta coefficient) for gender:
      The LOGISTIC Procedure

                            Analysis of Maximum Likelihood Estimates

                                              Standard          Wald

               Parameter    DF    Estimate       Error    Chi-Square    Pr > ChiSq

               Intercept     1      0.2179      0.0389       31.4605        <.0001

               IsFemale      1      0.6073      0.0639       90.3529        <.0001

To adjust for confounding by program, simply add program to the model. 
We can do this by directly modifying the automatically generated code. Select the Program tab. Scroll to find the PROC LOGISTIC code:

(1) In PROC LOGISTIC, replace the temporary dataset that SAS creates (using PROC SQL) with the original dataset. (The temporary dataset will not contain the variable program).  
PROC LOGISTIC data=admissions;
(2) Add Program to the model after isFemale (separated only by spaces).

PROC LOGISTIC… 


MODEL Denied (Event = '1')=

isfemale program/

Click Run to run the code:

(3) Say yes when it asks you if you want to modify the code and yes when it asks you if you want to replace the code.
Now we have a logistic model with two predictors, gender and program. The resulting beta coefficients are each “adjusted for” each other:

  Analysis of Maximum Likelihood Estimates

                                               Standard          Wald

              Parameter      DF    Estimate       Error    Chi-Square    Pr > ChiSq

              Intercept       1      0.6893      0.0513      180.2711        <.0001

              IsFemale        1     -0.0990      0.0809        1.4986        0.2209

11. We can also check for an interaction (effect modification) between gender and program by adding an interaction term into the model (directly modify the code again):

proc logistic data=admissions;

class program;

model denied (event="1") = isfemale program isfemale*program;

weight count;

run;

     Analysis of Maximum Likelihood Estimates

                                                  Standard          Wald

          Parameter             DF    Estimate       Error    Chi-Square    Pr > ChiSq

          Intercept              1      0.6573      0.0547      144.4267        <.0001

          IsFemale               1     -0.1858      0.1107        2.8175        0.0932

          program          A     1     -1.1443      0.0801      203.9757        <.0001

          program          B     1     -1.1834      0.0899      173.0928        <.0001

          program          C     1     -0.1218      0.1086        1.2567        0.2623

          program          D     1     -0.0334      0.1010        0.1092        0.7410

          program          E     1      0.5490      0.1506       13.2884        0.0003

          IsFemale*program A     1     -0.8714      0.2414       13.0361        0.0003

          IsFemale*program B     1     -0.0419      0.3740        0.0126        0.9108

          IsFemale*program C     1      0.3107      0.1614        3.7039        0.0543

          IsFemale*program D     1      0.2311      0.1655        1.9496        0.1626

          IsFemale*program E     1     0.00157      0.2016        0.0001        0.9938

PartII: McNemar’s Test is used when you have paired or matched data, such as in a matched case-control study. The idea is: you analyze the data by pairs rather than by individuals. Only pairs that are discordant (i.e., cases and controls differ on exposure) provide useful information. Under the null hypothesis, there should be just as many discordant pairs where the case is exposed as pairs where the control is exposed. (We will discuss McNemar’s formally in class next Monday).

12. Start a new program (Program(New Program). Input the following matched dataset* and run McNemar’s Test, using the option “agree” in PROC FREQ (use code).

data the_pill;

input caseuse $ contruse $ n;

datalines;

Y Y 10

Y N 57

N Y 13

N N 95

;

proc freq data=the_pill order=data;

tables caseuse*contruse /agree;

weight n;

run;

	Table of caseuse by contruse

	 
	contruse
	Total

	
	Y
	N
	

	caseuse
	 
	10
	57
	67

	Y
	Frequency
	
	
	

	
	Percent
	5.71
	32.57
	38.29

	
	Row Pct
	14.93
	85.07
	 

	
	Col Pct
	43.48
	37.50
	 

	N
	Frequency
	13
	95
	108

	
	Percent
	7.43
	54.29
	61.71

	
	Row Pct
	12.04
	87.96
	 

	
	Col Pct
	56.52
	62.50
	 

	Total
	Frequency
	23
	152
	175

	
	Percent
	13.14
	86.86
	100.00


There are 57 discordant pairs where the case was exposed; and only 13 where the control was exposed. Under the null hypothesis, there should  be 70/2=35 pairs of each. Is 57/13 significantly different from 35/35?  YES.
                          Statistics for Table of caseuse by contruse

	McNemar's Test

	Statistic (S)
	27.6571

	DF
	1

	Pr > S
	<.0001


*Example from “A Handbook of Statistical Analyses using SAS” Der and Everitt.

13.  Next, we will explore the connection between Cochran-Mantel-Haenszel and McNemar’s Tests. To turn the paired, grouped data into a dataset that contains 1 observation for each individual, use the following code (add to the current program):



data the_pill2;  

set the_pill;

if caseuse='Y' and contruse='Y' then do;  


do id=1 to n;


isCase=1; 


UsePill=1;


output;


isCase=0; 


UsePill=1;


output;


end;

end;

if caseuse='Y' and contruse='N' then do;


do id=101 to (n+100);


isCase=1; 


UsePill=1;


output;


isCase=0; 


UsePill=0;


output;


end;

end;

if caseuse='N' and contruse='Y' then do;


do id=201 to (n+200);


isCase=1; 


UsePill=0;


output;


isCase=0; 


UsePill=1;


output;


end;

end;

if caseuse='N' and contruse='N' then do;


do id=301 to (n+300);


isCase=1; 


UsePill=0;


output;


isCase=0; 


UsePill=0;


output;


end;

end;

run;

14. To better understand what we’ve done here, look at the dataset by clicking on the “Output Dataset” tab.  

15. Add to our program the following code to run the CMH test on the stratified data, where each pair forms its own unique stratum.  

proc freq data=the_pill2;

tables id*isCase*UsePill/nocol norow nopercent cmh measures;

run;
16. In the Results tab, scroll down to any of the 175 2x2 tables. Click on anyone to see what it looks like. Then Scroll to the very bottom to the “Summary for isCase*UsePill” to see the CMH statistics.

For example, table 53:

	Table 53 of isCase by UsePill

	Controlling for id=143

	 
	UsePill
	Total

	
	0
	1
	

	isCase
	 
	1
	0
	1

	0
	Frequency
	
	
	

	1
	Frequency
	0
	1
	1

	Total
	Frequency
	1
	1
	2


17. Compare the CMH chi-square statistic to the McNemar’s chi-square statistic (from step12):

	Cochran-Mantel-Haenszel Statistics (Based on Table Scores)

	Statistic
	Alternative Hypothesis
	DF
	Value
	Prob

	1
	Nonzero Correlation
	1
	27.6571
	<.0001

	2
	Row Mean Scores Differ
	1
	27.6571
	<.0001

	3
	General Association
	1
	27.6571
	<.0001





This is the risk of being admitted, rather than the risk of being denied (reduced 32% in females). 














Fisher’s Exact test is automatically generated for 2x2 tables; use when expected cell sizes < 5. Use the option “exact” or “Fisher” to ask for Fisher’s exact test for contingency tables larger than 2x2.








There is a significant interaction between program A and gender. 





To evaluate confounding, look at what happens to the beta coefficient for gender.


Adding program to the model removes the effect for gender completely (the beta coefficient for gender is now about 0).





There is a significant relationship between gender and admissions; the magnitude of the effect is reflected in a beta coefficient of +.60.





Adjusted RR=.95 (.87-1.03), suggesting that after controlling for program, women and men have about the same overall admissions rate.








Non-significant Cochran-Mantel Haenzel test tells us that after controlling for program, there’s no evidence of an association between gender and admissions. 





It is not necessary to use Fisher’s exact test here. If you were using Fisher’s exact test, you would report the two-sided probability.





Type a dollar sign after the variable name to make the variable a character variable. Numeric variables are the default.











Breslow-Day test actually rejects here because, unlike the other programs, women actually have an advantage in program A. So program A has a different relationship than the others, and we do not meet the assumption of homogeneity.





It’s really tough to get into program F, and not too hard to get into A or B.











SAS distinguishes between numeric variables (which can be manipulated arithmetically) and character variables.  Here, the first variable is character; the rest are numeric.











Programs A and B are heavily male; C and E are more female-dominated.





Note the use of the do loop and the if-then-do loop.  Here, a do-loop is nested inside an if-then-do loop.























Program and gender are strongly related.





Interaction term is the product of the two variables.

















Calculation of Chi-Square statistic


� EMBED Equation.3  ���





Asks for (Cochran) Mantel-Haenszel chi-square test and summary odds and risk ratios.








Crude OR 


Female vs. male: 1.84





Crude RR (females vs. males) =1.25 (1.20,1.31)





Conclusion: Women 25% more likely to be denied admission (and we’re 95% certain that the true increase in risk is somewhere between 20% and 31%).





Asks for crude OR and RR’s





n represents the number of pairs.





Strata variable*Row variable* Column Variable





I’m indexing id from 101 here rather than 1 to make sure that each case-control pair will have a unique, identifying id number (to keep the pairing).  











Note: values of the character variable are enclosed in single quotation marks.








output tells SAS to output one observation into the dataset being created; 


in the first iteration of the do loop here, the observation created by this output will have values of:


caseuse: N   


contruse: Y


n: 13


id: 201


isCase: 1


UsePill: 0





in the first iteration of the do loop here, the observation created by this output will have values of:


caseuse: N   


contruse: Y


n: 13


id: 201


isCase: 0


UsePill: 1











Inner loop:


Start by giving the variable id a value of 1.  Then go through the loop until end; then return to the beginning of the loop and increase id to 2; then go through the loop until end; then return to the beginning of the loop and increase id to 3…when id>n, proceed out of the loop.





Note: I name the indexing variable “id” here because I am using id to give each case-control pair a unique ID number.











Outer loop:


For all observations  where caseuse=Y and contruse=Y, do everything until end;


(there’s only 1 observation that meets these criteria: Y Y 10 because we have grouped data).





Remember, we have grouped data, so we can’t forget to weight by our count variable!





This is a regression model where gender is the predictor and denial of admission is the outcome.








Stratum-specific RR for Program A.





Asks for stratum-specific measures of association.














Option=agree





Ask for chi-square test





Suppresses annoying row, column, and overall percents…





If you care to see expected counts in the table











Strong evidence of an association between gender and admissions (p<.0001)

